The Importance of Being Formal II

Martin Henz

July 1, 2013

Traditional Logic

1 Complement

The terms that we encountered so far consist of single words such as `Greeks`. We call such terms *primitive*. Since terms describe subsets S of a given universe U, it is natural to provide an operation on terms that selects the complement of S, denoted as U/S. For this, we introduce an operation `non` that is applied to a `Term` and results in a `Term`. Thus, if our universe consists of all living things, then the `Term`

```
non Greeks
```

denotes those living things that are not Greeks. Note that the operator `non` is applied to a `Term` and returns a `Term`. Terms that are constructed using such an operator are called *compound* terms.

Of course, for any compound term, we can define a primitive term that is equal to it. For example, the following definition introduces a primitive term `immortal`.

Definition 1 (ImmortalDef)**.** *The term `immortal` is considered equal to the term `non mortal`.*

```
Parameter non: Term -> Term.
```

We can apply the `non` operation to a term, for example to `mortal`:

```
Definition immortal: Term := non mortal.

Definition SomeHumansAreImmortal : CategoricalProposition :=
   Some humans are immortal.
```

Since the complement of the complement of any set is the set itself, we are compelled to assert the following axiom.

Axiom 1 (NonNon)**.** *For any term t, the term* `non non` *t is considered equal to t.*

```
Axiom NonNon: forall t, non (non t) = t.
```

In graphical notation, the axiom can be depicted as follows.

$$\frac{\cdots t\ \cdots}{\cdots \texttt{non non}\ t \cdots}\,[\text{NNI}]$$

$$\frac{\cdots \texttt{non non}\ t\ \cdots}{\cdots t \cdots}\,[\text{NNE}]$$

where the $\cdots$ allow for any context. Here is an example instance of the first rule.

$$\frac{\texttt{Some}\ t_1\ \texttt{are}\ t_2}{\texttt{Some non non}\ t_1\ \texttt{are}\ t_2}$$

In these rules, the proposition above the bar states premises, and the formula below the bar states the conclusion. We can use the last rule to prove `Some non non` t'_1 `are` t'_2; then we are left with the obligation to prove `Some` t'_1 `are` t'_2.

The following lemma is proven with the use of this axiom, in addition to the previous axiom `HumansMortality`.

Lemma 1. *The proposition* `All humans are non immortal` *holds.*

We use our graphical notation to write down the proof for this lemma:

Proof.

$$\dfrac{\dfrac{\dfrac{}{\texttt{All humans are mortal}}\,[\text{HumansMortality}]}{\texttt{All humans are non non mortal}}\,[\text{NNI}]}{\texttt{All humans are non immortal}}\,[\text{ImmortalDef}]$$

□

When the proofs get larger, they often do not fit on a single page, and become difficult to draw. Therefore, we introduce an alternative text-based notation for proofs, where each proposition is given in a numbered line, and the rule that justifies the proposition is given on the right. For example, the text-based notation for the proof above reads as follows.

Proof.

1	`All humans are mortal`	HumansMortality
2	`All humans are non non mortal`	NNI 1
3	`All humans are non immortal`	ImmortalDef 2

□

The third column justifies the proposition, citing the axiom or definition used, if necessary referring to a previous line using its number.

```
Lemma HumansNonImmortality: All humans are non immortal.

Proof.
unfold immortal.
rewrite NonNon.
apply HumansMortality.
Qed.
```

Note that the directive `unfold` unfolds the definition of `immortal`, as given above. Now we find a double application of `non`, which presents us with an opportunity to rewrite the proposition using the equality asserted by the axiom `NonNon`. Finally, an application of `HumansMortality` completes the proof.

2 Immediate Inferences

This section introduces transformation operations on propositions, whose validity we intend to study.

2.1 Conversion

Consider the operation `convert`, that switches subject and object of a proposition. Thus we can define:

Definition 2 (ConvDef)**.** *For all terms t_1 and t_2, we define*

$$\begin{aligned}
\mathtt{convert}(\mathtt{All}\ t_1\ \mathtt{are}\ t_2) &= \mathtt{All}\ t_2\ \mathtt{are}\ t_1 \\
\mathtt{convert}(\mathtt{Some}\ t_1\ \mathtt{are}\ t_2) &= \mathtt{Some}\ t_2\ \mathtt{are}\ t_1 \\
\mathtt{convert}(\mathtt{No}\ t_1\ \mathtt{are}\ t_2) &= \mathtt{No}\ t_2\ \mathtt{are}\ t_1 \\
\mathtt{convert}(\mathtt{Some}\ t_1\ \mathtt{are\ not}\ t_2) &= \mathtt{Some}\ t_2\ \mathtt{are\ not}\ t_1
\end{aligned}$$

Note that not all conversions preserve the meaning of propositions; if `All Greeks are humans` holds in a model, then `All humans are Greeks` may or may not hold in the same model.

Conversion does preserve the meaning of particular affirmative and universal negative propositions, as expressed by the following axioms.

Axiom 2 (ConvE1)**.** *If, for some terms t_1 and t_2, the proposition*

$$\texttt{convert}(\texttt{Some } t_1 \texttt{ are } t_2)$$

holds, then the proposition

$$\texttt{Some } t_1 \texttt{ are } t_2$$

also holds.

Axiom 3 (ConvE2)**.** *If, for some terms t_1 and t_2, the proposition*

$$\texttt{convert}(\texttt{No } t_1 \texttt{ are } t_2)$$

holds, then the proposition

$$\texttt{No } t_1 \texttt{ are } t_2$$

also holds.

In graphical notation, two rules correspond to the two cases.

$$\frac{\mathit{convert}(\texttt{Some } t_1 \texttt{ are } t_2)}{\texttt{Some } t_1 \texttt{ are } t_2}\,[\mathrm{ConvE}_1]$$

$$\frac{\mathit{convert}(\texttt{No } t_1 \texttt{ are } t_2)}{\texttt{No } t_1 \texttt{ are } t_2}\,[\mathrm{ConvE}_2]$$

Exercise 1. *Use Venn diagrams and the definitions in Section 2 in* Traditional Logic I *to argue why Axioms 2 and 3 are valid.*

In Coq, the two axioms are formulated as follows.

```
Axiom ConvE1:
   forall subject object: Term,
   (Some object are subject)
   ->
   (Some subject are object).

Axiom ConvE2:
   forall subject object: Term,
   (No object are subject)
   ->
   (No subject are object).
```

As an example of conversion in action, consider the following axiom.

Axiom 4 (SomeAnimalsAreCats)**.** *The proposition* `Some animals are cats` *holds.*

```
Axiom SomeAnimalsAreCats : Some animals are cats.
```

We can use conversion, to prove `Some cats are animals` as follows.

Lemma 2. *The proposition* `Some cats are animals` *holds.*

Proof.

$$\dfrac{\dfrac{\dfrac{}{\texttt{Some animals are cats}}\,[\text{SomeAnimalsAreCats}]}{\texttt{convert(Some cats are animals)}}\,[\text{ConvDef}]}{\texttt{Some cats are animals}}\,[\text{ConvE}_1]$$

□

The same proof follows in text-based notation.

Proof.

1	`Some animals are cats`	SomeAnimalsAreCats
2	`convert(Some cats are animals)`	ConvDef 1
3	`Some cats are animals`	ConvE_1 2

□

Here is the Coq version of this proof.

```
Lemma SomeCatsAreAnimals : Some cats are animals.

Proof.
apply ConvE1.
apply SomeAnimalsAreCats.
Qed.
```

Similarly, the validity of conversion of universal negative propositions is used in the proof below.

Axiom 5 (NoGreeksAreCats)**.** *The proposition* `No Greeks are cats` *holds.*

We can use conversion, to prove `No cats are Greeks` as follows.

Lemma 3 (NoCatsAreGreeks)**.** *The proposition* `No cats are Greeks` *holds.*

Proof.

$$\dfrac{\dfrac{\dfrac{}{\texttt{No Greeks are cats}}\,[\text{NoGreeksAreCats}]}{\texttt{convert(No cats are Greeks)}}\,[\text{ConvDef}]}{\texttt{No cats are Greeks}}\,[\text{ConvE}_2]$$

□

In text-based notation, the proof goes as follows:

Proof.

1	`No Greeks are cats`	NoGreeksAreCats
2	`convert(No cats are Greeks)`	ConvDef 1
3	`No cats are Greeks`	ConvE_2 2

□

And, for completeness, the same in Coq:

```
Axiom NoGreeksAreCats : No Greeks are cats.

Lemma NoCatsAreGreeks : No cats are Greeks.

Proof.
apply ConvE2.
apply NoGreeksAreCats.
Qed.
```

2.2 Contraposition

Similar to conversion, we define an operation called *contraposition*, which reverses the roles of subject and object and forms the complement of both. Thus, the contraposition of `All Greeks are humans` is `All non humans are non Greeks`.

Definition 3 (ContrDef)**.** *For all terms t_1 and t_2, we define*

$$\begin{aligned}
\mathit{contrapose}(\texttt{All}\ t_1\ \texttt{are}\ t_2) &= \texttt{All non}\ t_2\ \texttt{are non}\ t_1\\
\mathit{contrapose}(\texttt{Some}\ t_1\ \texttt{are}\ t_2) &= \texttt{Some non}\ t_2\ \texttt{are non}\ t_1\\
\mathit{contrapose}(\texttt{No}\ t_1\ \texttt{are}\ t_2) &= \texttt{No non}\ t_2\ \texttt{are non}\ t_1\\
\mathit{contrapose}(\texttt{Some}\ t_1\ \texttt{are not}\ t_2) &= \texttt{Some non}\ t_2\ \texttt{are not non}\ t_1
\end{aligned}$$

Contraposition is a valid operation on universal affirmative and particular negative propositions.

Axiom 6 (ContrE1)**.** *If, for some terms t_1 and t_2, the proposition*

$$\mathit{contrapose}(\texttt{All}\ t_1\ \texttt{are}\ t_2)$$

holds, then the proposition

$$\texttt{All}\ t_1\ \texttt{are}\ t_2$$

also holds.

Axiom 7 (ContrE2)**.** *If, for some terms t_1 and t_2, the proposition*

$$\mathit{contrapose}(\texttt{Some}\ t_1\ \texttt{are not}\ t_2)$$

holds, then the proposition

$$\texttt{Some}\ t_1\ \texttt{are not}\ t_2$$

also holds.

```
Axiom ContrE1:
   forall subject object: Term,
   All (non object) are (non subject)
   ->
   All subject are object.

Axiom ContrE2:
   forall subject object: Term,
   Some object are not subject
   ->
   Some subject are not object.
```

In graphical notation, two rules correspond to the two cases.

$$\frac{\mathit{contrapose}(\texttt{All}\ t_1\ \texttt{are}\ t_2)}{\texttt{All}\ t_1\ \texttt{are}\ t_2}\,[\mathrm{ContrE}_1]$$

$$\frac{contrapose(\texttt{Some } t_1 \texttt{ are not } t_2)}{\texttt{Some } t_1 \texttt{ are not } t_2}\,[\mathrm{ContrE}_2]$$

Exercise 2. *Use contraposition to prove*

`All (non animals) are (non cats)`

using `All cats are animals` as axiom. Give your proof in graphical and text-based notation.

Sometimes you may not want to prove a lemma immediately. In Coq, you can write `admit` to "pretend" the current goal to be proven. Of course, proofs that contain `admit` are not valid. We use `admit` in order to pose exercises.

Coq Exercise 1. *Replace `admit` in the following to obtain a valid proof.*

```
Axiom AllCatsAreAnimals : All cats are animals.

Lemma AllNonAnimalsAreNonCats : All (non animals) are (non cats).

Proof.
admit.
Qed.
```

Hint: You may need to use `Axiom NonNon` from Section 1.

Exercise 3. *As an example for particular negative propositions, use contraposition to prove `Some non cats are not non animals` with the help of an axiom `Some animals are not cats`. Give your proof in graphical and text-based notation.*

Coq Exercise 2. *Replace `admit` in the following to obtain a valid proof.*

```
Axiom SomeAnimalsAreNotCats : Some animals are not cats.

Lemma SomeNonCatsAreNotNonAnimals : Some (non cats) are not (non animals).

Proof.
admit.
Qed.
```

2.3 Obversion

The final operation on propositions is called *obversion*. This operation complements the object and turns affirmative propositions into negative propositions, and vice versa. Thus, applying obversion to the universal affirmative proposition `All Greeks are humans` results in the universal negative proposition `No Greeks are non humans`, and the particular negative proposition `Some animals are not cats` results in the particular affirmative proposition `Some animals are non cats`.

Definition 4 (ObvDef)**.** *For all terms t_1 and t_2, we define*

$$\begin{aligned} \mathit{obvert}(\mathit{All}\ t_1\ \mathit{are}\ t_2) &= \mathit{No}\ t_1\ \mathit{are}\ \mathit{non}\ t_2 \\ \mathit{obvert}(\mathit{Some}\ t_1\ \mathit{are}\ t_2) &= \mathit{Some}\ t_1\ \mathit{are}\ \mathit{not}\ \mathit{non}\ t_2 \\ \mathit{obvert}(\mathit{No}\ t_1\ \mathit{are}\ t_2) &= \mathit{All}\ t_1\ \mathit{are}\ \mathit{non}\ t_2 \\ \mathit{obvert}(\mathit{Some}\ t_1\ \mathit{are}\ \mathit{not}\ t_2) &= \mathit{Some}\ t_1\ \mathit{are}\ \mathit{non}\ t_2 \end{aligned}$$

Obversion is valid for all kinds of propositions.

Axiom 8 (ObvE)**.** *If, for some proposition p*

$$\mathit{obvert}(p)$$

holds, then the proposition p also holds.

```
Axiom ObvE1:
   forall subject object: Term,
   All subject are non object
   ->
   No subject are object.

Axiom ObvE2:
   forall subject object: Term,
   No subject are non object
   ->
   All subject are object.


Axiom ObvE3:
   forall subject object: Term,
   Some subject are non object
   ->
   Some subject are not object.

Axiom ObvE4:
   forall subject object: Term,
   Some subject are not non object
   ->
   Some subject are object.
```

In our rule notation, we write the axiom as depicted below:

$$\dfrac{obvert(p)}{p}\,[\mathrm{ObvE}]$$

Example 1. *We prove* `Some humans are not non vegetarians` *from* `Some humans are vegetarians`*.*

Axiom 9 (SomeHumansVegetarians)**.** *The proposition* `Some humans are vegetarians` *holds.*

Lemma 4 (NNVeg)**.** *The proposition* `Some humans are not non vegetarians` *holds.*

Proof.

$$\dfrac{\dfrac{\dfrac{\dfrac{}{\texttt{Some humans are vegetarians}}\,[\text{SomeHumansVegetarians}]}{\texttt{Some humans are non non vegetarians}}\,[\text{NNI}]}{\texttt{obvert(Some humans are not non vegetarians)}}\,[\text{ObvDef}]}{\texttt{Some humans are not non vegetarians}}\,[\text{ObvE}]$$

□

We see here already that the graphical notation becomes inconvenient when proofs reach a certain size. The text-based alternative is more suitable for larger proofs.

Proof.

1	`Some humans are vegetarians`	SomeHumansVegetarians	
2	`Some humans are non non vegetarians`	NNI 1	
3	`obvert(Some humans are not non vegetarians)`	ObvDef 2	□
4	`Some humans are not non vegetarians`	ObvE 3	

In Coq, we use `rewrite NonNon`, as shown before.

```
Parameter vegetarians : Term.

Axiom SomeHumansVegetarians : holds (Some humans are vegetarians).

Lemma SomeHumansAreNotNonVegetarians :
   holds (Some humans are not (non vegetarians)).

Proof.
apply ObvE3.
rewrite NonNon.
apply SomeHumansVegetarians.
Qed.
```

2.4 Combinations

We can combine our three operations—conversion, contraposition and obversion—to obtain further results.

Lemma 5 (SomeNon)**.** *For all terms t_1 and t_2, if the proposition `Some non` t_1 `are non` t_2 holds, then the proposition `Some non` t_2 `are not` t_1 also holds.*

A lemma of the form "If p_1 then p_2" is valid, if in every model in which the proposition p_1 holds, the proposition p_2 also holds.

Exercise 4. *Argue that Lemma 5 is valid, using the definitions of the meaning of the respective propositions (Section 4 in* Traditional Logic I*). Use a Venn diagram to illustrate your argument.*

In order to prove a lemma of the form `If` p_1, `then` p_2, we need to obtain p_2, using p_1 as a *premise.*

Proof.

$$
\dfrac{\dfrac{\dfrac{\dfrac{\dfrac{}{\texttt{Some non } t_1 \texttt{ are non } t_2}\,[\text{premise}]}{\texttt{convert(Some non } t_2 \texttt{ are non } t_1\texttt{)}}\,[\text{ConvDef}]}{\texttt{Some non } t_2 \texttt{ are non } t_1}\,[\text{ConvE}_1]}{\texttt{obvert(Some non } t_2 \texttt{ are not } t_1\texttt{)}}\,[\text{ObvDef}]}{\texttt{Some non } t_2 \texttt{ are not } t_1}\,[\text{ObvE}]
$$

□

Proof.

1	`Some non` t_1 `are non` t_2	premise
2	`convert(Some non` t_2 `are non` t_1`)`	ConvDef 1
3	`Some non` t_2 `are non` t_1	ConvE_1 2
4	`obvert(Some non` t_2 `are not` t_1`)`	ObvDef 3
5	`Some non` t_2 `are not` t_1	ObvE 4

□

Note that a premise plays the same role as an axiom in a prove; it can be stated without a need for further proof.

```
Lemma SomeNon: forall subject object,
     Some non object are non subject
    -> Some non subject are not object.

Proof.
intros.
apply ObvE3.
apply ConvE1.
apply H.
Qed.
```

Note that the tactic `intros` transforms the original form p_1 `->` p_2 such that p_2 remains to be proven (below the bar), where p_1 can be assumed as a premise (above the bar).

3 Arguments and Syllogisms

An argument consists of a number of propositions, called premises, and a further proposition, called conclusion. For example, Lemma 5 is an argument with one premise, namely `Some non` t_1 `are non` t_2, whose conclusion is `Some non` t_2 `are not` t_1. An argument is valid if in any model in which all premises hold, the conclusion also holds.

A syllogism is an argument with two premises, in which three different terms occur, and in which every term occurs twice, but never twice in the same proposition.

Exercise 5. *Verify that Example 2 in* Traditional Logic I *contains a syllogism. Show that this syllogism is not valid, using a model as counter-example.*

The following syllogism consists of only universal affirmative propositions.

Axiom 10 (Barbara)**.** *For all terms minor, middle, and major, if `All` middle `are` major holds, and `All` minor `are` middle holds, then `All` minor `are` major also holds.*

The name Barbara for this syllogism is a mnemotic device used for remembering syllogisms. Traditionally, universal affirmative propositions are denoted by the letter 'a', universal negative propositions by the letter 'e', particular affirmative propositions by the letter 'i', and particular negative propositions by the letter 'o'. The vowels in Barbara indicate that all three propositions are universal affirmative.

The Barbara syllogism is graphically depicted as follows:

$$\frac{\texttt{All}\ \mathit{middle}\ \texttt{are}\ \mathit{major} \qquad \texttt{All}\ \mathit{minor}\ \texttt{are}\ \mathit{middle}}{\texttt{All}\ \mathit{minor}\ \texttt{are}\ \mathit{major}}\ [\text{Barbara}]$$

We convince ourselves that Barbara is valid, using the Venn diagram below.



If the first premise holds, then areas 1 and 4 are empty, and if the second premise holds, then areas 2 and 3 are empty. The conclusion simply states that areas 1 and 2 are empty, which clearly follows from the premises, regardless what other properties the model under consideration has.

We can directly apply Barbara for proving `All Greeks are mortal` from Axioms 2 and 3 in *Traditional Logic II*.

Lemma 6. *The proposition* `All Greeks are mortal` *holds.*

Proof.

$$\dfrac{\texttt{All Greeks are humans} \qquad \texttt{All humans are mortal}}{\texttt{All Greeks are mortal}}\,[\text{Barbara}]$$

□

In text-based form, the proof reads as follows.

Proof.

1	`All Greeks are humans`	GreeksHumanity
2	`All humans are mortal`	HumansMortality
3	`All Greeks are mortal`	Barbara 1,2

□

Since Barbara has two premises, the justification in Line 3 refers to two other lines, namely Line 1 and Line 2.

In Coq, we are formulating Barbara as an `Axiom` as follows.

```
Axiom Barbara : forall major minor middle: Thing -> Prop,
    All middle are major
 /\ All minor are middle
 -> All minor are major.

Lemma GreeksMortality : All Greeks are mortal.

Proof.
apply Barbara with (middle := humans).
split.
apply HumansMortality.
apply GreeksHumanity.
Qed.
```

Note that in order to apply `Barbara`, we need to tell Coq what term to use as `middle`. We use the tactic `split` in order to prove the two premises (put together by the conjunction symbol /\) separately.

4 Fun With Barbara

The English author and logician Lewis Carroll formulated many logical puzzles that can be solved using term logic. Our first example uses the following three premises.

- No ducks waltz.
- No officers ever decline to waltz.
- All my poultry are ducks.

Our version of the puzzle consists of proving from these premises that no officers are my poultry.

Lemma 7 (No-Officers-Are-My-Poutry)**.** *If `No ducks are things-that-waltz` holds, `No officers are non things-that-waltz` holds, and `All my-poutry are ducks` holds, then `No officers are my-poultry` also holds.*

Proof.

1	`No officers are non things-that-waltz`	premise
2	`obvert(All officers are things-that-waltz)`	ObvDef 1
3	`All officers are things-that-waltz)`	ObvE 2
4	`No ducks are things-that-waltz)`	premise
5	`convert(No things-that-waltz are ducks)`	ConvDef 4
6	`No things-that-waltz are ducks`	ConvE_2 5
7	`No things-that-waltz are non non ducks`	NNI 6
8	`obvert(All things-that-waltz are non ducks)`	ObvDef 7
9	`All things-that-waltz are non ducks`	ObvE 8
10	`All my-poultry are ducks`	premise
11	`All my-poultry are non non ducks`	NNI 10
12	`All non non my-poultry are non non ducks`	NNI 11
13	`contrapose(All non ducks are non my-poultry)`	ContrDef 12
14	`All non ducks are non my-poultry`	ContrE_1 13
15	`All things-that-waltz are non my-poultry`	Barbara 9,14
16	`All officers are non my-poultry`	Barbara 3,15
17	`obvert(No officers are my-poultry)`	ObvDef 16
18	`No officers are my-poultry`	ObvE 17

□

We first introduce suitable terms.

```
Parameter ducks : Term.
Parameter things_that_waltz : Term.
Parameter officers : Term.
Parameter my_poultry : Term.
```

The lemma and the proof follow.

```
Lemma No_Officers_Are_My_Poulty :
   holds (No ducks are things_that_waltz) /\
   holds (No officers are non things_that_waltz) /\
   holds (All my_poultry are ducks)
   ->
   holds (No officers are my_poultry).
```

```
Proof.
intro.
destruct H.
destruct H0.
apply ObvE1.
apply Barbara with (middle := things_that_waltz).
split.
apply Barbara with (middle := non ducks).
split.
apply ContrE1.
rewrite NonNon.
rewrite NonNon.
apply H1.
apply ObvE2.
rewrite NonNon.
apply ConvE2.
apply H.
apply ObvE2.
apply H0.
Qed.
```

Exercise 6. *From the following premises*

- *All babies are illogical.*
- *Nobody is despised who can manage a crocodile.*
- *Illogical persons are dispised.*

prove that no babies can manage a crocodile.

Coq Exercise 3.

```
Parameter babies : Term.
Parameter logical_persons : Term.
Parameter despised_persons : Term.
Parameter persons_who_can_manage_a_crocodile : Term.

Lemma No_baby_can_manage_a_crocodile :
All babies are non logical_persons /\
No persons_who_can_manage_a_crocodile are despised_persons /\
All non logical_persons are despised_persons
->
No babies are persons_who_can_manage_a_crocodile.
```

Complete the proof below.

```
Proof.
admit.
Qed.
```

Exercise 7. *From the following premises*

- *No boys under 12 are admitted to this school as boarders.*
- *All the industrious boys have red hair.*
- *None of the dayboys learn Greek.*
- *None but those under 12 are idle.*

prove that all boys who learn Greek are red-haired.

Coq Exercise 4. *Complete the proof in the following.*

```
Parameter boys_under_12 : Term.
Parameter boys_admitted_as_boarders : Term.
Parameter industrious_boys : Term.
Parameter red_haired_boys : Term.
Parameter boys_who_learn_Greek : Term.

Lemma All_Boys_Who_learn_Greek_are_red_haired :
   No boys_under_12 are boys_admitted_as_boarders /\
   All industrious_boys are red_haired_boys /\
   No non boys_admitted_as_boarders are boys_who_learn_Greek /\
   No non boys_under_12 are non industrious_boys
   ->
   All boys_who_learn_Greek are red_haired_boys.



Proof.
admit.
Qed.
```

Exercise 8. *From the following premises*

- *No interesting poems are unpopular among people of real taste.*
- *No modern poetry is free from affectation.*
- *All your poems are on the subject of soap-bubbles.*
- *No affected poetry is popular among people of real taste.*
- *No ancient poem is on the subject of soap-bubbles.*

prove that All your poems are non-interesting poems.

Exercise 9. *From the following premises*

- *The only animals in this house are cats.*
- *Every animal is suitable for a pet, that loves to gaze at the moon.*
- *When I detest an animal, I avoid it.*
- *No animals are carnivourous, unless they prowl at night.*
- *No cat fails to kill mice.*
- *No animals ever take to me, except what are in this house.*
- *Kangaroos are not suitable for pets.*
- *None but carnivora kill mice.*
- *I detest animals that do not take to me.*
- *Animals, that prowl at night, always love to gaze at the moon.*

prove that I avoid kangaroos.

5 Other Syllogisms

The syllogism under consideration in the previous sections only handles universal affirmative propositions. The following syllogism

Axiom 11 (Celarent[1])**.** *For all terms minor, middle, and major, if `No` middle `are` major holds, and `All` minor `are` middle holds, then `No` minor `are` major also holds.*

Exercise 10. *Use a Venn diagram to convince yourself of the validity of Celarent.*

Axiom 12 (Darii)**.** *For all terms minor, middle, and major, if `All` middle `are` major holds, and `Some` minor `are` middle holds, then `Some` minor `are` major also holds.*

Exercise 11. *Use a Venn diagram to convince yourself of the validity of Darii.*

The three syllogisms Barbara, Celarent and Darii have the following structure in common: The first premise shares its object with the conclusion, the second premise shares its subject with the conclusion, and no complement is used in any proposition. Such syllogisms are called *Figure 1* syllogisms.

[1]The vowels 'e', 'a' and 'e' in the word "Celarent" indicate that the first premise and the conclusion are universal negative ('e') and the second premise is universal affirmative ('a').

Exercise 12. *How many Figure 1 syllogisms (valid or not) exist?*

Exercise 13. *In addition to Barbara, Celarent and Darii, there is one other valid Figure 1 syllogism. Can you find it? Give it a name, according to the mnemonic device used for the other three.*